

[Description  
Options](#)[Quick start  
Remarks and examples](#)[Menu  
Also see](#)[Syntax](#)

## Description

graph twoway lpolyci plots a local polynomial smooth of *yvar* on *xvar* by using graph twoway line (see [G-2] [graph twoway line](#)), along with a confidence interval by using graph twoway rarea (see [G-2] [graph twoway rarea](#)).

## Quick start

Local polynomial smooth plot of *y* versus *x* with 95% confidence interval and local mean smoothing

```
twoway lpolyci y x
```

Same as above, but with a polynomial of degree 3

```
twoway lpolyci y x, degree(3)
```

Overlay scatterplot of the observed data on polynomial smooth plot

```
twoway lpolyci y x || scatter y x
```

Same as above, but with a 99% confidence interval

```
twoway lpolyci y x, level(99) || scatter y x
```

Specify the Gaussian kernel function

```
twoway lpolyci y x, kernel(gaussian) || scatter y x
```

Specify the half-width of the kernel as 85

```
twoway lpolyci y x, bwidth(85) || scatter y x
```

With a separate graph area for each level of *catvar*

```
twoway lpolyci y x || scatter y x, by(catvar)
```

Add “My Title” as the title of the graph

```
twoway lpolyci y x || scatter y x, title("My Title")
```

Draw the confidence limits as a pair of lines rather than as an area

```
twoway lpolyci y x, ciplot(rline) || scatter y x
```

## Menu

Graphics > Two-way graph (scatter, line, etc.)

Syntax

```
twoway lpolyci yvar xvar [if] [in] [weight] [, options]
```

options	Description
<u>kernel</u> ( <i>kernel</i> )	kernel function; default is kernel(epanechnikov)
<u>bwidth</u> (#)	kernel bandwidth
<u>degree</u> (#)	degree of the polynomial smooth; default is degree(0)
<u>n</u> (#)	obtain the smooth at # points; default is min( $N$ , 50)
<u>level</u> (#)	set confidence level; default is level(95)
<u>pwidth</u> (#)	pilot bandwidth for standard error calculation
<u>var</u> (#)	estimate of the constant conditional variance
<u>nofit</u>	do not plot the smooth
<u>fitplot</u> ( <i>plotype</i> )	how to plot the smooth; default is fitplot(line)
<u>ciplot</u> ( <i>plotype</i> )	how to plot CIs; default is ciplot(rarea)
<i>fcline_options</i>	change look of the smoothed line
<i>fitarea_options</i>	change look of CI
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

kernel	Description
<u>epanechnikov</u>	Epanechnikov kernel function; the default
<u>epan2</u>	alternative Epanechnikov kernel function
<u>biweight</u>	biweight kernel function
<u>cosine</u>	cosine trace kernel function
<u>gaussian</u>	Gaussian kernel function
<u>parzen</u>	Parzen kernel function
<u>rectangle</u>	rectangle kernel function
<u>triangle</u>	triangle kernel function

fweights and aweights are allowed; see [U] 11.1.6 weight.

Options

- kernel** (*kernel*) specifies the kernel function for use in calculating the weighted local polynomial estimate. The default is kernel(epanechnikov). See [R] **kdensity** for more information on this option.
- bwidth** (#) specifies the half-width of the kernel, the width of the smoothing window around each point. If bwidth() is not specified, a rule-of-thumb bandwidth estimator is calculated and used; see [R] **lpoly**.
- degree** (#) specifies the degree of the polynomial to be used in the smoothing. The default is degree(0), meaning local mean smoothing.
- n** (#) specifies the number of points at which the smooth is to be evaluated. The default is min( $N$ , 50), where  $N$  is the number of observations.

`level(#)` specifies the confidence level, as a percentage, for confidence intervals. The default is `level(95)` or as set by `set level`; see [U] 20.8 Specifying the width of confidence intervals.

`pwidth(#)` specifies the pilot bandwidth to be used for standard error computations. The default is chosen to be 1.5 times the value of the rule-of-thumb bandwidth selector.

`var(#)` specifies an estimate of a constant conditional variance required for standard error computation. By default, the conditional variance at each smoothing point is estimated by the normalized weighted residual sum of squares obtained from locally fitting a polynomial of order  $p+2$ , where  $p$  is the degree specified in `degree()`.

`nofit` prevents the smooth from being plotted.

`fitplot(plottype)` specifies how the prediction is to be plotted. The default is `fitplot(line)`, meaning that the smooth will be plotted by `graph twoway line`. See [G-2] graph twoway for a list of *plottype* choices. You may choose any that expects one  $y$  and one  $x$  variable. `fitplot()` is seldom used.

`ciplot(plottype)` specifies how the confidence interval is to be plotted. The default is `ciplot(rarea)`, meaning that the confidence bounds will be plotted by `graph twoway rarea`.

A reasonable alternative is `ciplot(rline)`, which will substitute lines around the smooth for shading. See [G-2] graph twoway for a list of *plottype* choices. You may choose any that expects two  $y$  variables and one  $x$  variable.

*fcline\_options* specify how the lpoly line is rendered and its appearance; see [G-3] fcline\_options.

*fitarea\_options* specify how the confidence interval is rendered; see [G-3] fitarea\_options. If you specify `ciplot()`, you should specify whatever is appropriate instead of using *fitarea\_options*.

*axis\_choice\_options* associate the plot with a particular  $y$  or  $x$  axis on the graph; see [G-3] axis\_choice\_options.

*twoway\_options* are a set of common options supported by all twoway graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] twoway\_options.

## Remarks and examples

`graph twoway lpoly ci yvar xvar` uses the `lpoly` command—see [R] lpoly—to obtain a local polynomial smooth of  $yvar$  on  $xvar$  and confidence intervals and uses `graph twoway line` and `graph twoway rarea` to plot results.

Remarks are presented under the following headings:


*Typical use*  
*Use with by()*

## Typical use

graph twoway lpolyci can be used to overlay the confidence bands obtained from different local polynomial smooths. For example, for local mean and local cubic polynomial smooths:

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)

. twoway lpolyci weight length, nofit ||
    lpolyci weight length, degree(3) nofit
    ciplot(rline) lcolor(gs6) ||
    scatter weight length, msymbol(o)
```



The plotted area corresponds to the confidence bands for the local mean smooth and lines correspond to confidence intervals for the local cubic smooth.


When you overlay graphs, you nearly always need to respecify the axis titles by using the *axis\_title\_options* `ytitle()` and `xtitle()`; see [\[G-3\] axis\\_title\\_options](#).

## Use with by()

graph twoway lpolyci may be used with by():

```
. use https://www.stata-press.com/data/r19/auto, clear
(1978 automobile data)

. twoway lpolyci weight length ||
    scatter weight length, msymbol(o) ||
    , by(foreign)
```



## Also see

[R] [lpoly](#) — Kernel-weighted local polynomial smoothing

[G-2] [graph twoway lpoly](#) — Local polynomial smooth plots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

